

How to Implement a Trinomial Option Pricing Model in MS-Excel?

Amit K. Sinha
Bradley University

Among the several popular option pricings models exists the binomial and trinomial option pricing modes. Several authors have shown how to easily implement in MS-Excel the binomial option pricing model which assumes prices to either go up or down only. The trinomial option pricing model, however, assumes prices could go up, down or remain the same. Although it is a viable and robust option pricing model, a simple procedure to implement it in Excel is not easy to find. This manuscript briefly discusses the trinomial option model, and using =IF() commands and coding in appropriate formulae, elucidates a simple procedure to implement it in Excel. The stock lattice structure can be easily extended to the desired size by clicking and dragging the cursor across, and then down. Using terminal period pay-offs, and recursively carrying out present value calculations, one period back, one at a time, option prices can be estimated for American and European options.

Keywords: trinomial option pricing model, MS - Excel

INTRODUCTION

The binomial option pricing model has been a very popular model for pricing options. It has witnessed several seminal papers. (Broadie & Detemple, 1996; Cox, Ross, & Rubinstein, 1979; Rendleman & Barter, 1979) are a few that have witnessed numerous citations. In the binomial model, stock price can jump either up or down. Building on binomial model, the trinomial model was developed, in which the stock price follows a three-jump process. It could jump up or down, or remain the same, at each point wherever the possibility of jump exists. Several manuscripts have also developed this model. (Boyle, 1986, 1988; Boyle & Tian, 1998; Rubinstein, 2000; Yuen & Yang, 2010) and (Hull, 2018) are a few that have discussed this model. This manuscript is not about developing a new trinomial model approach or identifying its shortcomings or strengths, but rather it shows how to implement a trinomial model in MS-Excel.

A few studies have looked at implementing the trinomial model. (Rouah & Vainberg, 2007) implement a trinomial model in MS-Excel but they use visual basic (VBA) codes. (Clifford, Wang, Aboronski, & Zhang, 2010) provide Matlab codes to implement a trinomial model. In a series of YouTube videos Brian Byrne (Byrne, 2018a, 2018b, 2018c) shows how to develop trinomial models using C++ and VBA codes. This manuscript differs from those manuscripts as it shows how to develop a trinomial lattice structure in MS-Excel that does not require any type of programming beyond using =IF() statements. The =IF() statements are used to develop the stock price lattice and estimate the price of American and European call and put options. The idea of using =IF() statements comes from (Holden, 2009) who used such statements to price binomial options. Section II of this manuscript discusses the trinomial model, while section III shows how to implement the trinomial model in MS-Excel. Section IV is the conclusion.

TRINOMIAL MODEL

The trinomial model discussed in this manuscript is from (Boyle, 1988) and (Hull, 2018). Consider the returns of a stock, S , to be log normally distributed. It flows a three-point jump process over a small interval of time, and the variance of its discrete distribution is equal to the variance of its lognormal distribution. The following notations are used: T is the time to option expiration in years; X is the exercise price of option; r is the continuously compounded annual return; r_f is the risk-free rate and assumed to be 4%; σ^2 is the yearly variance of the returns of the stock; n is the number of steps in which T is divided; h estimated as T/n is the length of one step.

FIGURE 1
TRINOMIAL STOCK PRICE LATTICE




Figure 1 depicts the lattice of a stock whose price follows a three-jump process. The initial stock price is S_0 , which could go up to S_0u , go down to S_0d , or remain as S_0 at the end of the first step. Once the prices jump to S_0u , it can jump three ways. It can jump up to S_0u^2 , remain at S_0u , or fall to S_0 . Similarly, if the prices were to fall to S_0d , it has three possibilities. It could fall further to S_0d^2 , remain at S_0d , or jump up to S_0 . If, however, prices were to move horizontally to S_0 , it could jump up to S_0u or fall down to S_0d . Similar possibilities exist after each jump.

(Boyle, 1988) estimates the size of the upstate using the formula in equation 1. This is the same as that developed by the (Cox et al., 1979), except that (Cox et al., 1979) assumed λ to be equal to one. (Boyle, 1988) assumed it to be greater than one. Also, $u^*d = 1$, where d is the size if prices fall.

$$u = e^{(\lambda\sigma\sqrt{h})} \tag{1}$$


(Boyle, 1988) used several values for λ , although (Hull, 2018) assumed it to be equal to square root of three. As is manuscript is an illustration, $\lambda = \sqrt{3}$ is used, which is the same as that used by (Hull, 2018). Let's assume that the stock has an annual standard deviation of 25%, and a return of 10%. The option on the stock expires in one year, and the stock price follows a three-step three jump process to the expiration of the options. Plugging in the numbers in equation 1, u and d are calculated as follows:

$$u = e^{(\sqrt{3} \cdot 25\% \cdot \sqrt{1/3})} = e^{(.25 \cdot \sqrt{3/3})} = e^{(.25)} = 1.2840 \tag{2}$$

$$d = \frac{1}{u} = \frac{1}{1.2840} = 0.7788 \tag{3}$$

Figure 2 shows the lattice of a stock with the initial price of thirty dollars, using the values of u and d estimated in equations 2 and 3.

**FIGURE 2
NUMERICAL EXAMPLE OF TRINOMIAL STOCK PRICE LATTICE**



The probability of prices going up is assumed to be p_1 , and that of going down, p_3 , while the probability for prices to remain the same is p_2 . The total of $p_1 + p_2 + p_3$ is equal to one. Boyle 1988) also provides solutions for p_1 , p_2 , and p_3 , but the solutions used in this manuscript is from (Hull, 2018), and are shown equations 4, 5, and 6. In these formulae q is the continuously compounded dividend yield, and has the value of zero for

non-dividend paying stocks. For the purpose of the illustration in this manuscript, a non-dividend paying stock is considered.

$$p_1 = \frac{1}{6} + \sqrt{\frac{h}{12\sigma^2}} \left(r - q - \frac{\sigma^2}{2} \right) \quad (4)$$

$$p_2 = \frac{2}{3} \quad (5)$$

$$p_3 = \frac{1}{6} - \sqrt{\frac{h}{12\sigma^2}} \left(r - q - \frac{\sigma^2}{2} \right) \quad (6)$$

Plugging in values in equation 4 and 6, the probability of prices going up or down is obtained in equation 7 and 9.


$$p_1 = \frac{1}{6} + \sqrt{\frac{\frac{1}{3}}{12 * 0.25^2}} \left(.1 - 0 - \frac{0.25^2}{2} \right) = 0.2125 \quad (7)$$

$$p_2 = \frac{2}{3} = 0.667 \quad (8)$$

$$p_3 = \frac{1}{6} - \sqrt{\frac{\frac{1}{3}}{12 * 0.25^2}} \left(.1 - 0 - \frac{0.25^2}{2} \right) = 0.1208 \quad (9)$$

The price of the option is based on calculating the pay-offs at each node. The expected value of those pay-offs is discounted back on step back to the previous jump. For example, let's look at Figure 3 which corresponds to the top-right of the lattice structure depicted in Figure 2. The pay-offs of the call option is first estimated at expiration using the traditional call option pay-off formulae: $C = \text{Max}(0, S - X)$. The pay-offs are represented by cf_1 , cf_2 , and cf_3 .

**FIGURE 3
ONE STEP BACK VALUATION**



$$cf_3 = \text{Max}(0, S_0u^3 - X)$$


$$cf_2 = \text{Max}(0, S_0u^2 - X)$$

$$cf_1 = \text{Max}(0, S_0u - X)$$

Figure 4 shows the pay-offs at each node and the estimated call-option price. Consider the portion of the lattice circled. The pay-offs at expiration for the European call, and the probabilities calculated in equations 7, 8 and 9 are used to estimate the call value at the node pointed out by the dashed arrow, and is shown in equation 10 below. By repeating the calculations back to the starting point the price of the call option can be calculated as 4.45.

$$c = e^{-\left(\frac{1}{3}\right)*0.04}(33.51 * 0.2125 + 19.46 * 0.6667 + 8.52 * 0.1208) = 20.84 \quad (10)$$


FIGURE 4
EUROPEAN CALL OPTION PRICE



Similarly, using the same stock price lattice shown in Figure 2, and the probabilities from equations 7, 8 and 9, the price of an European put can be estimated. The pay-off's at expiration are estimated using the traditional put option pay-off formulae: $P = \text{Max}(0, X - S)$. Figure 5 show pay-offs and the value at each node and price of the European put with a strike price of 30 to be 1.44. The calculation at the node being pointed out by the dashed arrow in Figure 5, is based on the calculation shown in equation 11.

$$p = e^{-\left(\frac{1}{3}\right)*0.04}(6.64 * 0.2125 + 11.80 * 0.6667 + 15.83 * 0.1208) = 11.04 \quad (11)$$

FIGURE 5
EUROPEAN PUT OPTION PRICE



The American call and put option prices can be also estimated and are shown in Figures 6 and 7 respectively. Similar to the European options, the first step is to estimate the pay-off at expiration. Then the value at the nodes preceding the expiration has to be obtained, and at these nodes, the calculation has to be adjusted such that it allows for early exercise of the option. For example, consider the node in Figure 6, that is pointed out by the dashed arrow. The present value of the pay-offs is the same as that in equation 10. However, equation 10 does not allow for early exercise of the option. Equation 12 is modified equation 10 that allows for early exercise of the option. In this equation, 30 is the strike price, while 49.46 is the stock price, at the respective node. The calculations at each of the nodes have to be adjusted accordingly.

$$c = \text{Max} \left(49.46 - 30, e^{-\left(\frac{1}{3}\right) \cdot 0.04} (33.51 * 0.2125 + 19.46 * 0.6667 + 8.52 * 0.1208) \right) = 20.84 \quad (12)$$


The value at each node is calculated similarly by considering the values of three subsequent nodes and the pay-offs of early expiration. Repeating this calculation recursively one step back, the price of the American call can be estimated at time = 0. In the example considered here, it happens that the American call price is the same as that of the European call.

The American put option price can be similarly calculated, by suitably adjusting the European put option. For example, the equation 11, has been adjusted to equation 13 for the estimating the value at the node being pointed out by the dashed arrow in Figure 7.


$$p = \text{Max} \left(30 - 18.20, e^{-\left(\frac{1}{3}\right) * 0.04} (6.64 * 0.2125 + 11.80 * 0.6667 + 15.83 * 0.1208) \right) = 11.80 \quad (13)$$

If prices were to follow a path that it ends up at 18.20, then the American put with the exercise price of 30 would be exercised at this node. Similar to the call option calculations, the price of the American put, can be estimated by recursive calculations one step back at a time, till time zero. In the example being considered, the American put is priced higher than the European put.

**FIGURE 6
AMERICAN CALL**



**FIGURE 7
AMERICAN PUT**



TRINOMIAL MODEL IN EXCEL

This section shows how to implement the trinomial model using MS-Excel functionalities. It starts with the basic calculations of the size jumps up and down, as well as those of the probabilities going up or down. Figure 8 shows these calculations along with the Excel functions that were used. The Excel function =FORMULATEXT() are used to show the formula or function used in the cell adjacent and on the left. The =ROUND() function has been used. It is not necessary to use that but has been used to be consistent with the numbers arrived at in the previous section.

**FIGURE 8
BASIC CALCULATIONS IN EXCEL**

		B12 : ✕ ✓ fx =ROUND(EXP(B11*SQRT(3*B7)),4)	
	A	B	C
1	Initial Stock Price (S ₀)	30	
2	Strike Price (X)	30	
3	Annual Stock return	10%	
4			
5	Expiration of Option	1	
6	Number of steps	3	
7	Time for one step	0.3333	=1/B6
8	annual return	10%	=B3
9	Risk Free Return	4%	
10	Dividend Yield (q)	0%	
11	Standard Deviation	25%	
12	Size of upstate (U)	128.40%	=ROUND(EXP(B11*SQRT(3*B7)),4)
13	midstate	100%	
14	Size of downstate (D)	77.88%	=ROUND(1/B12,4)
15	Dividend Yield (q)	0%	
16	p ₁	0.2125	=ROUND(1/6+((SQRT(B7/(12*B11^2)))*(B8-B15-(B11^2/2))),4)
17	p ₂	0.6667	=ROUND(2/3,4)
18	p ₃	0.1208	=ROUND(1/6-((SQRT(B7/(12*B11^2)))*(B8-B15-(B11^2/2))),4)
19		1.000	=B16+B17+B18

The calculations are used by appropriately linking the cells to develop the stock price lattice structure. Figure 9A shows the upper and lower halves of the lattice, while Figure 9B combines them together into a single trinomial stock price lattice. In Figure 9A, cell F8 is linked to cell B1 which contains the initial stock price, while cell F10 is linked to F8 to serve as the initial price for the lower half of the lattice. The formula typed in cell G8 is shown in cell E4. Once the formula has been typed out, the cursor was placed on cell G8. Thereafter, it was allowed to hover at the bottom right of that cell till the '+' sign appears. Once the '+' appears, the cursor can be clicked to drag across to cell I8, and then released. This highlights G8:I8. Keeping those cells highlighted, and hovering bottom right in cell I8, brings the '+' once again. This time around, the cursor is dragged up to cell I5 to produce the upper half of the lattice structure. The bottom half of the lattice structure is obtained by first typing out the formula shown in cell E14 in G10, and dragging the cursor across and down over the appropriate cells.

**FIGURE 9A
UPPER AND LOWER STOCK PRICE LATTICE**

	E	F	G	H	I
1	Step	0	1	2	3
2	Time	0	0.333	0.667	1.000
3					
4	=IF(F8="",IF(F9="", "", F9*\$B\$12),F8*\$B\$13)				
5					63.51
6	=B1			49.46	49.46
7			38.52	38.52	38.52
8		30.00	30.00	30.00	30.00
9					
10		30.00	30.00	30.00	30.00
11			23.36	23.36	23.36
12	=F8			18.20	18.20
13					14.17
14	=IF(F10="",IF(F9="", "", F9*\$B\$14),F10*\$B\$13)				

The formula typed in cell F25 in Figure 9B is shown in cell E22. This formula is dragged across to cell I25. The formula typed in cell F24 is shown in cell E20. This is dragged across to I24 and up to I22. The formula typed in cell F26 is shown in cell E30. This is dragged across to I26 and down to I28. Once complete, the trinomial lattice structure is obtained. The numbers in the structure is similar to in Figure 2. It should be noted that Figures 9A and 9B are both parts of the same worksheet in an Excel spreadsheet, with Figure 9B below 9A at the appropriately numbered row.

**FIGURE 9B
TRINOMIAL STOCK PRICE LATTICE**

	E	F	G	H	I
17	Step	0	1	2	3
18	Time	0	0.333	0.667	1.000
19					
20	=IF(F7="", "", F7)				
21					
22	=F8				63.51
23				49.46	49.46
24			38.52	38.52	38.52
25		30.00	30.00	30.00	30.00
26			23.36	23.36	23.36
27				18.20	18.20
28					14.17
29					
30	=IF(F11="", "", F11)				

Once the stock price lattice shown in Figure 9B is obtained, it can be used to estimate the pay-offs of the European put at expiration. Figure 10A shows the calculation. The call pay-off formula $C = \text{Max}(0, S_t - X)$ is typed in cell I37 in Figure 10A as $\text{Max}(0, I22 - \$B\$2)$. Once this formula is dragged down to cell I43, all possible pay-offs at expiration are obtained. Thereafter the formula shown show in cell E35, is typed out in cell H37. Once that formula is dragged down to H43 the column of numbers as observed in cells H38:H42 would be obtained. Once the formulae in cells H37:H43 are dragged back and across to F37:F43, the price of the European call option would have been calculated and it would appear in cell F40 as shown in Figure 10A. The logic in the $=\text{IF}()$ commands ensures that only the relevant pay-offs are considered when the value at the previous node is estimated. The $=\text{IF}()$ also presents a clean trinomial lattice structure.

FIGURE 10A
EUROPEAN CALL OPTION PRICE

	E	F	G	H	I	J	K
32	Step	0	1	2	3		
33	Time	0	0.333	0.667	1.000		
34							
35	$=\text{IF}(I36="", "", \text{IF}(I38="", "", (I36*\$B\$16+I37*\$B\$17+I38*\$B\$18)*\text{EXP}(-\$B\$7*\$B\$9)))$						
36							
37					33.51	$=\text{MAX}(0, I22 - \$B\$2)$	
38				20.84	19.46		
39			10.96	9.69	8.52		
40		4.45	3.21	1.79	0.00		
41			0.37	0.00	0.00		
42				0.00	0.00		
43					0.00		

Similarly, the European put option price can be calculated as shown in Figure 10B. Cells I50:I56 shows the pay-offs for the put-option at expiration using the put option pay-off formula. The formula typed in cell H50 is shown in cell E48. This formula can be copied and pasted over cells F50:H56 to obtain the put option price in cell F53. As before, the $=\text{IF}()$ command produces the trinomial lattice structure.

**FIGURE 10B
EUROPEAN PUT OPTION PRICE**

	E	F	G	H	I	J	K
45	Step	0	1	2	3		
46	Time	0	0.333	0.667	1.000		
47							
48	=IF(I49="", "", IF(I51="", "", (I49*\$B\$16+I50*\$B\$17+I51*\$B\$18)*EXP(-						
49	\$B\$7*\$B\$9)))						
50					0.00	=MAX(0,\$B\$2-I22)	
51				0.00	0.00		
52			0.09	0.00	0.00		
53		1.44	1.21	0.79	0.00		
54			5.28	5.77	6.64		
55				11.04	11.80		
56					15.83		

The price of the American options can be estimated just as easily as that of the European options. Figure 11A shows the calculations related to American call. It starts with estimating the pay-offs of the call pays at expiration and in Figure 11A they are calculated in cells I63:I69 using the pay-off formula for call options at expiration. The formula in cell H63 is adjusted to include the early exercise of the option by incorporating equation 12. This formula typed in cell E61. This formula can be copied and pasted over F63:H69. Alternatively, it could be dragged down and then across to obtain the American call option price in cell F66. The =IF() commands ensures that the calculation uses the trinomial lattice structure.

**FIGURE 11A
AMERICAN CALL OPTION PRICE**

	E	F	G	H	I	J	K
57							
58	Step	0	1	2	3		
59	Time	0	0.333	0.667	1.000		
60							
61	=IF(I62="", "", IF(I64="", "", MAX(H22-						
62	\$B\$2,(I62*\$B\$16+I63*\$B\$17+I64*\$B\$18)*EXP(-\$B\$7*\$B\$9)))						
63					33.51	=MAX(0,I22-\$B\$2)	
64				20.84	19.46		
65			10.96	9.69	8.52		
66		4.45	3.21	1.79	0.00		
67			0.37	0.00	0.00		
68				0.00	0.00		
69					0.00		

Figure 11B shows how the price of the American put option can be calculated in Excel. The put option pay-offs are estimated at expiration using the formula shown in cell J76. The formula in cell H76 is adjusted to allow for early expiration. This formula when copied and pasted over cells F76:H82 provides the price of the American put in cell F79.

FIGURE 11B
AMERICAN PUT OPTION PRICE

	E	F	G	H	I	J	K
71	Step	0	1	2	3		
72	Time	0	0.333	0.667	1.000		
73							
74	=IF(I75="", "", IF(I77="", "", MAX(\$B\$2-						
75	H22,(I75*\$B\$16+I76*\$B\$17+I77*\$B\$18)*EXP(-\$B\$7*\$B\$9))))						
76						0.00	=MAX(0,\$B\$2-I22)
77				0.00	0.00		
78			0.09	0.00	0.00		
79		1.67	1.31	0.79	0.00		
80			6.64	6.64	6.64		
81				11.80	11.80		
82					15.83		

CONCLUSION

Option pricing models are of continued interest in the Finance literature. Developing the tree based binomial option pricing model, a three-jump model the trinomial option model was developed. While there exists C++ programs and VBA codes to implement the trinomial model, in this manuscript, using =IF() statements a simple yet elegant approach to implement the trinomial model in MS-Excel has been developed.

REFERENCES

- Boyle, P.P. (1986). Option Valuation Using a Three-Jump Process. *International Options Journal*, 3, 5–12.
- Boyle, P.P. (1988). A Lattice Framework for Option Pricing With Two State Variables. *Journal of Financial and Quantitative Analysis*, 23(1), 1–12.
- Boyle, P.P., & Tian, Y. (Sam). (1998). An explicit finite difference approach to the pricing of barrier options. *Applied Mathematical Finance*, 5(1), 17–43. <https://doi.org/10.1080/135048698334718>
- Broadie, M., & Detemple, J. (1996). American Option Valuation: New Bounds, Approximations, and a Comparison of Existing Methods. *Review of Financial Studies*, 9(4), 1211–1250. <https://doi.org/10.1093/rfs/9.4.1211>
- Byrne, B. (2018a). *Trinomial Model 1*. Retrieved from <https://youtu.be/Vqy7h4XT1wc>
- Byrne, B. (2018b). *Trinomial Model 2*. Retrieved from https://youtu.be/wJnv7Y9v_3g
- Byrne, B. (2018c). *Trinomial Model 3*. Retrieved from <https://youtu.be/RYYI-0bc048>

- Clifford, P., Wang, Y., Aboronski, O., & Zhang, K. (2010). *Pricing Options Using Trinomial Trees*. Retrieved from https://warwick.ac.uk/fac/sci/math/people/staff/oleg_zaboronski/fm/trinomial_tree_2010_kevin.pdf
- Cox, J.C., Ross, S.A., & Rubinstein, M. (1979). Option pricing: A simplified approach. *Journal of Financial Economics*, 7(3), 229–263. [https://doi.org/10.1016/0304-405X\(79\)90015-1](https://doi.org/10.1016/0304-405X(79)90015-1)
- Holden, C.W. (2009). *Excel Modeling and Estimation in Investments* (3rd ed.). Upper Saddle River: Pearson Prentice Hall.
- Hull, J. (2018). *Options, Futures, and Other Derivatives* (10th ed.). Pearson.
- Rendleman, R.J., & Bartter, B.J. (1979). Two-State Option Pricing. *The Journal of Finance*, 34(5), 1093. <https://doi.org/10.2307/2327237>
- Rouah, F.D., & Vainberg, G. (2007). *Option Pricing Models and Volatility Using Excel-VBA*. Wiley Finance.
- Rubinstein, M. (2000). On the Relation Between Binomial and Trinomial Option Pricing Models. *The Journal of Derivatives*, 8(2), 47–50. <https://doi.org/10.3905/jod.2000.319149>
- Yuen, F.L., & Yang, H. (2010). Option pricing with regime switching by trinomial tree method. *Journal of Computational and Applied Mathematics*, 233(8), 1821–1833. <https://doi.org/10.1016/j.cam.2009.09.019>